

<input type="checkbox"/> Gr. 1, Dr. H. Dobler	Name _____	Aufwand in h _____
<input type="checkbox"/> Gr. 2, Dr. G. Kronberger		
<input type="checkbox"/> Gr. 3, Dr. H. Dobler	Punkte _____	Übungsleiter _____

Im Moodle-Kurs zu dieser Lehrveranstaltung finden Sie in der Datei *Coco-2.zip* die .NET-Version von Coco-2 (in und für C#).

## 1. MiniC: Scanner und Parser mit Coco-2

(8 Punkte)

MiniC kennen Sie ja schon. Zur Wiederholung: Unten links ist ein einfaches Programm zur Berechnung des Satzes von Pythagoras dargestellt, rechts die Grammatik von MiniC (die Sie auch im Moodle-Kurs in der Datei *MiniC.syn* finden):

```
void main() {
    int a, b, cs;
    scanf(a);
    scanf(b);
    cs = (a * a) + (b * b);
    printf(cs);
}
```

```
MC =      "void" "main" "(" ")" "{"
        [ VarDecl ]
        StatSeq
        "}" .
VarDecl = "int" ident { "," ident } ";" .
StatSeq = Stat { Stat } .
Stat =    [ ident "=" Expr
          | "scanf" "(" ident ")"
          | "printf" "(" Expr ")"
          ] ";" .
Expr =    Term { ( "+" | "-" ) Term } .
Term =    Fact { ( "*" | "/" ) Fact } .
Fact =    ident | number | "(" Expr ")" .
```

Erzeugen Sie mit Coco-2 einen lexikalischen Analysator (*scanner*) und einen Syntaxanalysator (*parser*) für MiniC und bauen Sie daraus ein Programm für die Analyse von MiniC-Programmen.

## 2. MiniCpp: Scanner, Parser und ... mit Coco-2

(16 Punkte)

MiniCpp kennen Sie zwar auch schon, hier aber trotzdem das Beispiel zur Wiederholung:

```
void Sieve(int n); // declaration

void main() {
    int n;
    cout << "n > ";
    cin >> n;
    if (n > 2)
        Sieve(n);
} // main

void Sieve(int n) { // definition
    int col, i, j;
    bool *sieve = 0;
    sieve = new bool[n + 1];
    i = 2;
    while (i <= n) {
        sieve[i] = true;
        i++;
    } // while
```

```
cout << 2 << " ";
col = 1;
i = 3;
while (i <= n) {
    if (sieve[i]) {
        if (col == 10) {
            cout << endl;
            col = 0;
        } // if
        col++;
        cout << i << " ";
        j = i * i;
        while (j <= n) {
            sieve[j] = false;
            j = j + 2 * i;
        } // while
    } // if
    i = i + 2;
} // while
delete[] sieve;
} // Sieve
```

Hier die Grammatik für MiniCpp, die Sie auch im Moodle-Kurs in der Datei *MiniCpp.syn* finden:

```

MiniCpp =      { ConstDecl | VarDef | FuncDecl | FuncDef } .
ConstDecl =    'const' Type ident Init ';' .
Init =         '=' ( false | true | number ) .
VarDef =       Type [ '*' ] ident [ Init ]
               { ',' [ '*' ] ident [ Init ] } ';' .

FuncDecl =     FuncHead ';' .
FuncDef =      FuncHead Block .
FuncHead =     Type [ '*' ] ident '(' [ FormParList ] ')' .
FormParList =  ( 'void' |
                 Type [ '*' ] ident [ '[' ']' ]
                 { ',' Type [ '*' ] ident [ '[' ']' ] } ) .
Type =         'void' | 'bool' | 'int' .
Block =        '{' { ConstDecl | VarDef | Stat } '}' .
Stat =         ( IncStat | DecStat | AssignStat
                 | CallStat | IfStat
                 | WhileStat | BreakStat
                 | InputStat | OutputStat | DeleteStat | ReturnStat
                 | Block
                 | ';'
                 ) .

IncStat =      ident '++' ';' .
DecStat =      ident '--' ';' .
AssignStat =   ident [ '[' Expr ']' ] '=' Expr ';' .
CallStat =     ident '(' [ ActParList ] ')' ';' .
ActParList =   Expr { ',' Expr } .
IfStat =       'if' '(' Expr ')' Stat [ 'else' Stat ] .
WhileStat =    'while' '(' Expr ')' Stat .
BreakStat =    'break' ';' .
InputStat =    'cin' '>>' ident ';' .
OutputStat =   'cout' '<<' ( Expr | string | 'endl' )
               { '<<' ( Expr | string | 'endl' ) } ';' .

DeleteStat =   'delete' '[' ']' ident ';' .
ReturnStat =   'return' [ Expr ] ';' .
Expr =         OrExpr .
OrExpr =       AndExpr { '|' AndExpr } .
AndExpr =      RelExpr { '&&' RelExpr } .
RelExpr =      SimpleExpr
               [ ( '==' | '!=' | '<' | '<=' | '>' | '>=' )
                 SimpleExpr ] .

SimpleExpr =   [ '+' | '-' ]
               Term { ( '+' | '-' ) Term } .
Term =         NotFact { ( '*' | '/' | '%' ) NotFact } .
NotFact =      [ '!' ] Fact .
Fact =         'false' | 'true'
               | number
               | ident [ ( '[' Expr ']' )
                       | ( '(' [ ActParList ] ')' )
                       ]
               | 'new' Type '[' Expr ']'
               | '(' Expr ')' .

```

Erzeugen Sie mit Coco-2 einen lexikalischen Analysator (*scanner*) und einen Syntaxanalysator (*parser*) für MiniCpp und ... bauen Sie daraus ein Programm für die Analyse von MiniCpp-Programmen. Von diesem Werkzeug für die *statische Programmanalyse* sollen mindestens

- die Anzahl der Zeilen (*lines of code*, LOC),
- die Anzahl der Anweisungen (*statements*) und
- die Strukturkomplexität  $V$  nach McCabe ( $V = 1 + \text{Anzahl der binären Verzweigungen}$ )

berechnet werden. Natürlich wäre auch  $EV$  (die essentielle Strukturkomplexität =  $V$  nach Reduktion der D-Diagrammanteile) von Interesse, vielleicht schaffen Sie das ja auch noch.

### 3. Für besonders Interessierte, deshalb ...

(+ 4 Bonuspunkte)

M. H. Halstead hat Metriken vorgeschlagen, die nicht nur die Größe eines Algorithmus (in Form seiner "Länge"  $N$ , s. u.) und seine Struktur (wie McCabe) in Betracht ziehen, sondern vor allem die Umfang und Komplexität der darin durchgeführten Berechnungen berücksichtigen. Dafür müssen die vier in Tab. 0.1 definierten Werte  $n1$ ,  $N1$ ,  $n2$  und  $N2$  aus dem Quelltext eines Algorithmus ermittelt werden.

	Anzahl unterschiedlicher ...	Gesamtanzahl der verwendeten ...
... Operatoren	$n1$	$N1$
... Operanden	$n2$	$N2$

Tab. 0.1: Werte zur Berechnung der Halstead-Metriken

Dabei gelten als Operatoren nicht nur die mathematischen Operatoren (z. B. die arithmetischen  $+$ ,  $-$ ,  $*$ ,  $/$  und die relationalen  $==$ ,  $<$ ,  $>$ ,  $...$ ), sondern auch Symbole (z. B.  $=$ ) sowie jene Schlüsselwörter (z. B. *if*, *else* und *while*), die Aktionen in einem Algorithmus auslösen. Als Operanden werden alle Elemente eines Algorithmus gewertet, die Daten repräsentieren (vor allem Literale, Konstanten, Variablen, aber auch Sprungmarken). Aus den vier Größen in Tab. 0.1 können die in Tab. 0.2 genannten fünf Halstead-Metriken berechnet werden.

Halstead-Metrik	Bezeichnung	Formel
Alphabet (engl. <i>vocabulary</i> )	$n$	$n = n1 + n2$
Länge (engl. <i>length</i> )	$N$	$N = N1 + N2$
Volumen (engl. <i>volume</i> )	$V$	$V = N \cdot \log_2(n)$
Schwierigkeit (engl. <i>difficulty</i> )	$D$	$D = (n1 \cdot N2) / (2 \cdot n2)$
Aufwand (engl. <i>effort</i> )	$E$	$E = D \cdot V$

Tab. 0.2: Halstead-Metriken und ihre Berechnung

Erweitern Sie Ihr Werkzeug zur statischen Analyse von MiniCpp-Programmen aus 2. um die Berechnung der oben erläuterten Halstead-Metriken.

## Übung 3

# 1 MiniC: *Scanner* und *Parser* mit Coco-2

Listing 1: MiniC.atg

```

1  /* Comments are allowed here. COMPILER name is the same name as the file */
2  COMPILER MiniC
3
4  SEM<<
5      private static String identStr;
6      private static int  numberVal;
7  >>
8
9  CHARACTER SETS
10     digit      = '0' .. '9'.
11     letter     = 'a' .. 'z' + 'A' .. 'Z' .
12     whiteSpace = CHR(9) + EOL IGNORE. /* ' ' ignored by default*/
13     noquote    = ANY - '"'.
14
15  KEYWORDS
16     'const'. 'void'. 'main'. 'return'. 'delete'. 'new'.
17     'int'. 'bool'. 'false'. 'true'.
18     'if'. 'else'. 'while'. 'break'.
19     'cin'. 'cout'. 'scanf'. 'printf'.
20
21  TOKENS
22     ';' . ',' .
23     '=' . '+' . '-' . '++' . '--' . '/' . '*' . '%'.
24     '|' . '&&' . '<' . '>' . '<<' . '>>' . '<=' . '>=' . '==' . '!=' . '!' .
25     '(' . ')' . '[' . ']' . '{' . '}' .
26
27  TOKEN CLASSES
28     ident << out String identStr >> =
29         letter {letter | digit} LEX << identStr = tokenStr; >>.
30     number << out int numberVal >> =
31         digit {digit} LEX << numberVal=Convert.ToInt32(tokenStr); >>.
32
33  NONTERMINALS
34     MiniC.
35     VarDecl.
36     StatSeq.
37     Stat.
38     Expr.
39     Term.
40     Fact.
41
42  RULES
43     MiniC = 'void' 'main' '(' ')' '{' [ VarDecl ] StatSeq '}'.
44     VarDecl = 'int' ident<< out identStr >> { ',' ident<< out identStr >> } ';'.
45     StatSeq = Stat { Stat } .
46     Stat = [ ident << out identStr >> '=' Expr
47             | 'scanf' '(' ident<< out identStr >> ')'
48             | 'printf' '(' Expr ')'
49             ] ';'.
50     Expr = Term { ( '+' | '-' ) Term } .
51     Term = Fact { ( '*' | '/' ) Fact } .
52     Fact = ident<< out identStr >> | number << out numberVal >> | '(' Expr ')' .
53
54  END MiniC.

```


Listing 2: C Testprogram

## Übung 3

```

1 void main() {
2     int a, b, cs;
3     scanf(a);
4     scanf(b);
5     cs = (a * a) + (b * b);
6     printf(cs);
7 }

```



```

PS D:\repositories\Github\FH-SE-Master\Formale-Sprachen-Compiler-und-Werkzeugbau\Uebung_5\src\Coco-2> .\SG.exe ..\MiniC\MiniC.atg
C o c o - 2 SG Version C#, 1.1
Copyright H. Dobler Sept. 2011
-----
source file in "..\MiniC\MiniC.atg"
parsing ...
evaluating lexical structure ...
generating scanner ...
listing ...


compilation completed in 0:0.58

no errors detected
PS D:\repositories\Github\FH-SE-Master\Formale-Sprachen-Compiler-und-Werkzeugbau\Uebung_5\src\Coco-2> .\PGT.exe ..\MiniC\MiniC.atg
C o c o - 2 PGT Version C#, 1.1
Copyright H. Dobler Sept. 2011
-----
source file in "..\MiniC\MiniC.atg"
parsing ...
generating semantic evaluator ...
evaluating syntax ...
checking grammar ...
generating syntax analyzer ...
listing ...

compilation completed in 0:00.79
no errors detected
PS D:\repositories\Github\FH-SE-Master\Formale-Sprachen-Compiler-und-Werkzeugbau\Uebung_5\src\Coco-2>

```

Abbildung 1: Konsolenausgabe des Erstellens des *Parser* und *Scanners*



```

D:\repositories\Github\FH-SE-Master\Formale-Sprachen-Compiler-und-Werkzeugbau\Uebung_5\src\MiniC>.\MiniC.exe
-----
MiniC Compiler Version X
Frontend gen. with Coco-2
-----

source file > test.c
parsing ...
no errors detected

[c]ontinue or [q]uit >

```

Abbildung 2: Konsolenausgabe des *Parser*

## 2 MiniCpp: *Scanner* und *Parser* mit Coco-2

Listing 3: MiniCPP.atg

```

1 /* Attributed grammar for MiniCPP */
2 COMPILER MiniCPP
3
4 SEM<<
5     private static int loc = 0;
6     private static int stc = 0;
7     private static int mc = 0;
8 >>

```

## Übung 3

```

9
10
11 CHARACTER SETS
12 digit    = '0' .. '9'.
13 letter   = 'a' .. 'z' + 'A' .. 'Z' .
14 whiteSpace = CHR(9) + EOL IGNORE. /*' ' ignored by default*/
15 noquote  = ANY - '"'.
16
17 COMMENTS
18 FROM '//' TO EOL.
19 FROM '/*' TO '*/'.
20
21 KEYWORDS
22 'void'. 'new'. 'return'. 'const'. 'break'.
23 'int'. 'false'. 'true'. 'bool'.
24 'scanf'. 'printf'. 'cin'. 'cout'. 'endl'.
25 'if'. 'while'. 'else'. 'for'. 'delete'.
26
27 TOKENS
28 ';'. ','.
29 '=', '+', '-', '++', '--', '/', '*', '%'.
30 '|', '&&', '<', '>', '<<', '>>', '<=', '>=', '==', '!=', '!'.
31 '(', ')', '[', ']', '{', '}'.
32
33 TOKEN CLASSES
34 number = digit { digit }.
35 ident = ( letter | '_' ) { letter | digit | '_' }.
36 string = '"' { noquote } '"'.
37
38 NONTERMINALS
39 MiniCPP.
40 ConstDecl.
41 Init.
42 VarDef.
43 FuncHead.
44 FormParList.
45 Type.
46 Block.
47 Stat.
48 Expr.
49 Term.
50 Fact.
51 ActParList.
52 IfStat.
53 WhileStat.
54 BreakStat.
55 InputStat.
56 OutputStat.
57 DeleteStat.
58 ReturnStat.
59 OrExpr.
60 AndExpr.
61 RelExpr.
62 SimpleExpr.
63 NotFact.
64 VarDefOrFuncDeclOrDef.
65 FormParListRest.
66 NonVoidType.
67 IdentStat.
68
69 RULES
70
71 MiniCPP          = { ConstDecl | VarDefOrFuncDeclOrDef }

```

# Übung 3

```

72 SEM<<
73 Console.WriteLine($"Lines of code: {loc}");
74 Console.WriteLine($"Lines of statements: {stc}");
75 Console.WriteLine($"Complexity by McCabe: {mc}");
76 >>.
77 VarDefOrFuncDeclOrDef = Type [ '*' ] ident (VarDef | FuncHead ( ';' | Block ) SEM<<loc++;>>) .
78 ConstDecl = 'const' Type ident Init ';' SEM<<loc++;>> .
79 Init = '=' ( false | true | number ) .
80 VarDef = [ Init ] { ',' [ '*' ] ident [ Init ] } ';' SEM<<loc++;>>.
81 FuncHead = '(' [ FormParList ] ')' .
82 FormParList = ( 'void' [ FormParListRest ] | NonVoidType FormParListRest ) .
83 FormParListRest = [ '*' ] ident [ '[' ']' ] { ',' Type [ '*' ] ident [ '[' ']' ] } .
84 Type = 'void' | NonVoidType .
85 NonVoidType = 'bool' | 'int' .
86 Block = '{' { ConstDecl | Type [ '*' ] ident VarDef | Stat } '}'
  ↳ SEM<<loc++;>>.
87 Stat = ( IdentStat | IfStat | WhileStat | BreakStat | InputStat
88         | OutputStat | DeleteStat | ReturnStat | Block | '(' SEM<<stc++;>>)
  ↳ ).
89 IdentStat = ident ( '++' | '--' | [ '[' Expr ']' ] '=' Expr | '(' [ ActParList ]
  ↳ ')') ';' .
90 ActParList = Expr { ',' Expr } .
91 IfStat = 'if' '(' Expr ')' Stat [ 'else' Stat SEM<<mc++;>> SEM<<loc++;mc++;>>.
92 WhileStat = 'while' '(' Expr ')' Stat .
93 BreakStat = 'break' ';' .
94 InputStat = 'cin' '>>' ident ';' .
95 OutputStat = 'cout' '<<' ( Expr | string | 'endl' ) { '<<' ( Expr | string |
  ↳ 'endl' ) } ';' .
96 DeleteStat = 'delete' '[' ']' ident ';' .
97 ReturnStat = 'return' [ Expr ] ';' .
98 Expr = OrExpr .
99 OrExpr = AndExpr { '|' AndExpr } .
100 AndExpr = RelExpr { '&&' RelExpr } .
101 RelExpr = SimpleExpr [ ( '=' | '!=' | '<' | '<=' | '>' | '>=' ) SimpleExpr ]
  ↳ .
102 SimpleExpr = [ '+' | '-' ] Term { ( '+' | '-' ) Term } .
103 Term = NotFact { ( '*' | '/' | '%' ) NotFact } .
104 NotFact = [ '!' ] Fact .
105 Fact = 'false'
106         | 'true'
107         | number
108         | ident [ ( '[' Expr ']' )
109         | ( '(' [ ActParList ] ')' )
110         | 'new' Type '[' Expr ']'
111         | '(' Expr ')' .
112
113 END MiniCPP.

```

Listing 4: C++ Testprogram

```

1 void Sieve(int n); // declaration
2
3 void main() {
4     int n;
5     int x = 0;
6     cout << "n > ";
7     cin >> n;
8     if (n > 2)
9         Sieve(n);
10

```

## Übung 3

```

11     if(n > 2) {
12         x++;
13     }
14     else {
15         x--;
16     }
17
18
19 } // main
20
21 void Sieve(int n) { // definition
22     int col, i, j;
23     bool *sieve = 0;
24
25     sieve = new bool[n + 1];
26     i = 2;
27
28     while (i <= n) {
29         sieve[i] = true;
30         i++;
31     } // while
32
33     cout << 2 << " ";
34     col = 1;
35     i = 3;
36
37     while (i <= n) {
38         if (sieve[i]) {
39             if (col == 10) {
40                 cout << endl;
41                 col = 0;
42             } // if
43
44             col++;
45             cout << i << " ";
46             j = i * i;
47
48             while (j <= n) {
49                 sieve[j] = false;
50                 j = j + 2 * i;
51             } // while
52         } // if
53
54         i = i + 2;
55     } // while
56
57     delete[] sieve;
58 } // Sieve

```



## Übung 3

```

Windows PowerShell
PS D:\repositories\Github\FH-SE-Master\Formale-Sprachen-Compiler-und-Werkzeugbau\Uebung_5\src\Coco-2> .\SG.exe ..\MiniCPP\MiniCPP.atg
-----
C o c o - 2  SG      Version C#, 1.1
Copyright H. Dobler      Sept. 2011
-----

source file in "..\MiniCPP\MiniCPP.atg"
parsing ...
evaluating lexical structure ...
generating scanner ...
listing ...

compilation completed in 0:0.67

no errors detected
PS D:\repositories\Github\FH-SE-Master\Formale-Sprachen-Compiler-und-Werkzeugbau\Uebung_5\src\Coco-2> .\PGT.exe ..\MiniCPP\MiniCPP.atg
-----
C o c o - 2  PGT      Version C#, 1.1
Copyright H. Dobler      Sept. 2011
-----

source file in "..\MiniCPP\MiniCPP.atg"
parsing ...
generating semantic evaluator ...
evaluating syntax ...
checking grammar ...
listing ...
generating syntax analyzer ...

compilation completed in 0:00.85
LL(1) conflict(s) detected
PS D:\repositories\Github\FH-SE-Master\Formale-Sprachen-Compiler-und-Werkzeugbau\Uebung_5\src\Coco-2>

```

Abbildung 3: Konsolenausgabe des Erstellens des *Parser* und *Scanners*